

KING FAHD UNIVERSITY OF PETROLEUM AND MINERALS
DEPARTMENT OF ELECTRICAL ENGINEERING
Electronic Circuits II – EE303

Tutorial # 1
More About Computer Aided Circuit Analysis Using SPICE

In this tutorial an additional number of SPICE commands will be introduced and more circuits will be tested.

Control Statements

.TRAN

The inclusion of the statement .TRAN makes SPICE perform a transient analysis of the circuit over a specified period of time. The general form is

.TRAN TSTEP TSTOP

where TSTEP is the time increment used for plotting and/or printing results of the transient analysis; it is not necessarily the computing time step that SPICE uses between successive analyses. TSTOP is the time of the last transient analysis.

.OP

The inclusion of the statement .OP makes SPICE perform an analysis to find the operating point of the circuit with capacitors opened and inductors shorted, and to print detailed results of the operating point analysis. The general form is .OP

.TF

The inclusion of a .TF statement makes SPICE perform a small signal DC analysis yielding the transfer function between a specified output node and a specified input source. The output of the program can print the input resistance at the specified input source, the output resistance at the specified output node and transfer function. The general function is

.TF OUTPUTVAR INPUTSRC

where OUTPUTVAR is a small signal output variable (voltage or current) and INPUTSRC is a small signal source (voltage or current).

.PRINT

The inclusion of the .PRINT statement makes SPICE perform a print of a specified output variable resulting from a specified type of analysis. The general form is

.PRINT ANALYSISTYPE OV

where ANALYSISTYPE is the type of analysis performed from which the output variable OV is obtained.

Time Dependent Functions for Independent Sources

In SPICE there are five kinds of time-dependent functions for transient analysis using independent sources. Here we shall introduce only three of them.

PULSE

The general form is

V... N+ N- PULSE(V1 V2 TD TR TF PW PER)

I... N+ N- PULSE(V1 V2 TD TR TF PW PER)

Figure 1 shows a pulse form which illustrates the parameters mentioned above.




Figure 1

SIN

The general form is

V... N+ N- SIN(VO VA FREQ TD THETA)

I... N+ N- SIN(VO VA FREQ TD THETA)

Figure 2 shows a sinusoidal form which illustrates the parameters mentioned above.




Figure 2

PWL

The general form of piecewise linear function (PWL) is

$$V \dots N+ \quad N- \quad \text{PWL}(T1 \ V1 \ T2 \ V2 \ T3 \ V3 \ \dots)$$

$$I \dots N+ \quad N- \quad \text{PWL}(T1 \ V1 \ T2 \ V2 \ T3 \ V3 \ \dots)$$

Figure 3 shows a piecewise linear form which illustrates the parameters mentioned above.




Figure 3

PSPICE PROBE

Probe in PSPICE is the graphics postprocessor that calculates and displays results of a simulation. In effect, Probe functions as a “software” oscilloscope, calculator, and spectrum analyzer. Arithmetic operations on output variables are allowed in PSPICE Probe. The available operators and functions are as follows:

Function	Actual Meaning	Example
+	Addition of voltage or current	V(3)+V(5)
-	Subtraction of voltage or current	I(VSEN1)-I(VSEN2)
*	Multiplication of voltage or current	V(1)*I(VSEN1)
/	Division of voltage or current	V(10)/V(1)
ABS(X)	X-Absolute value of X	ABS(V(10))
SQRT(X)	X-Square root of X	SQRT(I(VSEN))
EXP(X)		EXP(V(5))
LOG(X)	ln(X) → log base e	LOG(V(1))
LOG10(X)	Log ₁₀ (X) → log base 10	LOG10(V(10))
DB(X)	20*log ₁₀ (X)	VDB(6)
PWR(X,Y)	X ^Y → X to the power of Y	PWR(V(5),2)
SIN(X)	Sin(X) → X in radians	SIN(V(1))
COS(X)	Cos(X)	COS(V(1))
TAN(X)	Tan(X)	TAN(V(1))
ATAN(X)	Atan(X)	ATAN(V(1))

D(X)	Derivative of X with respect to the x-axis variable d(V(3))	
S(X)	Integral of X over the x-axis variable	S(I(VSEN1))
AVG(X)	Running average of X	AVG(V(5))
RMS(X)	Running RMS average of X	RMS(I(VSEN10))

The general form of Probe is **.PROBE**, in this case all variables are available in the data file. Another form is **.PROBE V(1) V(4) V(7)**, in this case only voltage at nodes 1, 4 and 7 will be stored in the data file.

Examples

- The input file of the circuit of Figure 4 is:

```
PWRSUP.CIR
VAC      10 0 SIN(0 100 50)
D        10 11 MODA D
.MODEL  MODA D
CF       12 0 40U
RL       12 0 1K
RS       11 12 2
.TRAN    1M 40M
.PLOT   TRAN V(12)
.PROBE  V(12)
.END
```




Figure 4

- The input file of the circuit of Figure 5 is:

```
DIFF.CIR
RS1      1 2 500
RS2      6 0 500
VIN      1 0 AC 50M
RC1      11 3 10K
RC2      11 5 10K
VCC      11 0 15
VEE      9 0 -15
Q1       3 2 4 MOD1
Q2       5 6 4 MOD1
Q3       4 7 8 MOD1
R1       7 0 82K
R3       7 9 82K
R2       8 9 8.2K
.MODEL  MOD1 NPN
.TF    V(3) VIN
.END
```




Figure 5

3. The input file of the circuit of Figure 6 is:

```

TRANRC.CIR
VIN      5 0 PULSE(0 4 10N 2N 2N 20N 48N)
R        5 8 1K
C        8 0 400P
.TRAN 3N 105N
.PLOT TRAN V(8)
.PROBE
.END

```




Figure 6

At this stage you are requested to write and run the above three programs, obtain the results from SPICE simulation. Try to compare your results with your hand calculations using the approximate analysis techniques of EE 203.

We also encourage simulating and comparing with hand calculations for circuits of your own; try some of the circuits you studied in EE 203.